카일릭스 3를 이용한 쇼핑몰 구축 
내용
카일릭스가 가장 많이 사용되는 곳은 웹 애플리케이션 개발이며 또한 최근 쇼핑몰 구축에 많은 독자들이 관심을 가지고 있다. 이에 이번 카일릭스를 사용해 쇼핑몰을 구축하는 연재를 진행하려 한다. PHP나 ASP, JSP로도 다양한 쇼핑몰을 구축할 수 있지만 카일릭스를 사용하면 더 쉽고 빠르게 개발할 수 있는 장점이 있다. 볼랜드에 콤포넌트 기술과 강력한 DB 엔진, 그리고 IDE 환경으로 개발 사이클을 단축시키고 유지 보수와 코드 재사용을 보장할 수 있다. 

 
글 | 이갑연 ㈜네오아이시 사장(볼랜드포럼 카일릭스 채널 시삽) 

 

구성
1회, 카일릭스로 구축하는 쇼핑몰 소개 및 개괄적인 웹 애플리케이션 개발 방법 
    그리고 데이터베이스 연동 및 게시판 개발 실전 
2회, 쇼핑몰의 데이터베이스 설계(회원 등록 및 로그인/로그아웃) 

3회, 쇼핑몰의 검색 기능 구현(장바구니, 주문/발송조회, 관리자 시스템) 

지금까지 인트라웹을 이용하여 게시판과 회원등록까지 구현해보았다.

이제 마지막회로 쇼핑몰을 완성해보자. 쇼핑몰의 고차원적인 부분까지는 구현하지 못하더라도 기본이 되는  제품검색, 장바구니, 주문처리, 관리자기능까지는 구현해 보고자 한다.

구현할 기능

그림1은 쇼핑몰에 메인화면으로 로그인을 하면 세션에 로그인 정보를 저장한다.

전체적인 메뉴구성은 왼쪽에 Tframe을 사용하여 메뉴가 항상 동일하게 적용되도록 하였다.

그림1.메인화면

로그인이 완료가 되면 로그인창은 숨기고 쇼핑관련 메뉴를 보이게 한다.

[image: image2.png]2 EARIN®

| 13O BEE I EAWI® s2@ ssE® 0 &

EHE -9 - QA ) QM mENw @822 (B 5

(D) [ http://127.0,0.1:3414/EXEC/3/ 1CEDT30IE IDBAGCBAEBE4D

£173 4E1 B2 ElHotmail £]Windows &]22 Hotmail &]Windows Media &]index,asp-sve=

YELE

-ADs0R

o1

]

o
B
in

o

0 M

32 £ELEE H/B

oo

E N il
FHRLEDI

i3

a

NI





그림2.쇼핑

상품에 수량을 결정하여 장바구니에 담으면 장바구니에 추가되어 장바구니화면으로 이동한다. 장바구니화면에서는 상품수정과 삭제가 가능하고 주문처리도 이 화면에서 가능하다.(그림4)

그림3.장바구니화면

그림4.주문화면

주문이 완료되면 자기가 주문한 내역을 조회해보는 기능이다.

마지막으로 관리자기능으로 ID가 ADMIN이면 관리자로 인식하여 이화면으로 들어갈 수 있게 했다. 여기에서는 고객현황, 상품관리, 주문현황을 구현하면 된다. 

그림5.관리자화면

이제 실전으로 단계적으로 구현을 해보자

1.프래임으로 메뉴구성

이제 메뉴를 구성하는 프래임부터 만들어보자.

인트라웹은 프래임을 지원을 해주고 있다. 프래임을 사용하면 메뉴를 여러창에서 동일하게 보여줄수가 있다.

File->New->Frame 이렇게 해서 프래임을 추가시켜주고

[image: image7.png]



위와 같이 디자인을 한다.

그다음 모든폼에 콤포넌트 파렛트에서 Standard->Frames 를 폼에 올리고 방금만든 프래임을 연결하면 프래임만 수정하면 이프래임을 사용하는 폼들은 모두 자동 적용이 된다.

2.로그인 유지확인

로그인 유지는 로그인할 때 세션에 로그인 정보를 기억하고 있다. 그리고 폼이 이동될때마다 계속 유지되는지 확인하기 위해서는 프래임이 생성될때마다 체크를 하면 된다.

TMenuFrame = class(TFrame)

… 생략 …
public

    { Public declarations }

    constructor Create(AOwner: TComponent); override;

  end;
Crate를 오버로드해서 다음과 같이 코딩해준다.

즉 로그인이 되지 않았으면 로그인 영역이보이고 로그인이되면 로그인명하고 로그인 메뉴를 보여준다.

constructor TMenuFrame.Create(AOwner: TComponent);

begin

  inherited;

  //로그인 되었는가 확인

  if UserSession.membername <> '' then

  begin

    lUser.caption := UserSession.membername + '님';

    IWRegionLogout.Visible := true;

    IWRegionLogin.Visible  := False;

  end

  else

  begin

    IWRegionLogout.Visible := False;

    IWRegionLogin.Visible  := True;

  end;

end;
3.메뉴이동 구현하기

폼을 이동하기위해서는 데이터모듈에 공용함수인 move를 작성한다.

procedure Move(AFormClass: TIWAppFormClass);

begin

  // 현재폼 해제

  TIWAppForm(RWebApplication.ActiveForm).Release;

  // 새로운 폼 생성

  AFormClass.Create(RWebApplication).Show;

end;
이렇게 만든 함수는 아래와 같이 사용이 된다.

//회원가입

  move(TMemberRegForm);

  DataModuleForm.cdsMember.Append;

//쇼핑

  move(TShoppingForm);

//홈

  move(TMainForm);

//장바구니

  move(TCartForm);

//주문조회

  move(TOrderForm);

//관리자

  //ID가  ADMIN 이 관리자로 함.

  if UserSession.memberid <> 'ADMIN' Then Exit;

  move(TAdminForm);
4.상품리스트 출력하기

상품리스트는 이 앞전 만든 게시판과 비슷하나 이미지와 메모는 그리드에서 바로 표시되지는 않는다. 그래서 콤포넌트팔렛트에 IW Data->TIWDBImage 와 IW Data-> TIWDBMemo을 먼저 화면에 올려두고 datasource와 필드를 정의한다. 

IWDBGrid1->Columns에서 … 버튼 클릭하여 control에  각각컨트롤만 연결시키면 된다. 

5.상품검색하기

상품검색은 SQL문을 사용하여 검색하는 것이 보통이지만 지금여기서 구현하는 mybase는 filter 를 지원해준다. 필터를 사용하면 다양한 검색적용이 가능하다.

자세한 내용은 함수 레퍼런스를 참조하기 바란다.

지금은 상품설명에 있는내용을 가지고 검색하는 기능을 만들어 보겠다.

검색버튼 이벤트에 다음과 같이 코딩을 한다.

procedure TShoppingForm.btnSerClick(Sender: TObject);

begin

  //검색버튼

  DataModuleForm.cdsProduct.Filter := 'desc like ''%' + eSer.Text + '%''';

  DataModuleForm.cdsProduct.Filtered := eSer.Text <> '';

end;
이렇게 하면 내용중에 문장이 들어간 상품만 나열이 될것이다.

6.장바구니에 저장하기

상품을 선택하면 리스트 아래에 상품에 자세한 내역과 이미지가 나오는데 여기서 수량을 선택하고 장바구니 담기버튼을 클릭 하면 장바구니 데이터에 상품정보와 수량을 등록시키고 장바구니 화면으로 이동시키는 작업을 구현한다.

//장바구니 담기

  with  DataModuleForm.cdsCart do

  begin

    append;

    fieldbyname('memberid').asString := UserSession.memberid;

    fieldbyname('prodcode').asString := DataModuleForm.cdsProduct.fieldbyname('code').asString;

    fieldbyname('prodname').asString := DataModuleForm.cdsProduct.fieldbyname('name').asString;

    fieldbyname('saleprice').asString := DataModuleForm.cdsProduct.fieldbyname('saleprice').asString;

    fieldbyname('qty').asString := eQty.text;

    fieldbyname('amount').AsInteger := fieldbyname('saleprice').asinteger * fieldbyname('qty').asinteger;

    fieldbyname('regdate').asDatetime := Date;  //등록일자

    post;

    SaveToFile('./cart.xml');

  end;

  move(TCartForm);
7.장바구니수정

장바구니 수량수정은 리스트에서 수정한 상품을 클릭시 현재위치로 레코드를 현재위치로 이동시키고 eQty텍스트 박스에 값을 저장시킨다.

procedure TCartForm.IWDBGrid1Columns1Click(ASender: TObject;

  const AValue: String);

begin
//상품명 클릭시

  DataModuleForm.cdsCart.Locate('prodcode',AValue,[]);

  eQty.text :=DataModuleForm.cdsCart.fieldbyname('qty').asString;

end;
수정버튼 이벤트를 작업한다.

procedure TCartForm.IWButton1Click(Sender: TObject);

begin

//수량수정

  with DataModuleForm.cdsCart do

  begin

    edit;

    fieldbyname('qty').asString := eQty.text;

    fieldbyname('amount').AsInteger := fieldbyname('saleprice').asinteger * fieldbyname('qty').asinteger;

    fieldbyname('regdate').asDatetime := Date;  //등록일자

    post;

    SaveToFile('./cart.xml');

  end;

end;
8.주문폼으로 가기

주문처리는 장바구니 화면에서 아래쪽에 Region을 두어서 주문처리창 버튼을 클릭하면 

내용을 보이게 구현했다.

procedure TCartForm.IWButton3Click(Sender: TObject);

begin

//주문창열기

  IWRegionOrder.Visible := true;

end;
9.주문처리

주문처리는 내용입력을 받고 주문처리버튼을 클릭하면 Order 테이블과 OrderItem에 

장바구니 내용을 옮겨주고 장바구니는 비우는 기능을 수행한다.

주문처리 이벤트에 다음과 같이 코딩한다.

procedure TCartForm.IWButton4Click(Sender: TObject);

var

  iSum : integer; // 상품총액계산

  iQty : integer; //  상품건수

  sOrderNo : String;

  function LastOrderNo : String;

  var sDate :string;

      iCnt  :integer;

  begin

    sDate := formatdatetime('yyyymmdd',date);

    with DataModuleForm.cdsTemp do

    begin

      close;

      LoadFromFile('./order.xml');

      filter := 'orderno = ''' + sDate + '*''';

      filtered := true;

      iCnt := Recordcount + 1;

      close;

    end;

    Result := sDate + '-' + formatfloat('000',iCnt)

  end;

begin

//주문처리

  sOrderNo := LastOrderNo;

  with DataModuleForm do

  begin

    // 주문상품 저장

    iSum := 0;

    iQty := 0;

    while not cdsCart.Eof do

    begin

      cdsOrderItem.append;

      cdsOrderItem.fieldbyname('memberid').asString  := UserSession.memberid;

      cdsOrderItem.fieldbyname('orderno').asString   := sOrderNo;

      cdsOrderItem.fieldbyname('prodcode').asString  := cdsCart.fieldbyname('prodcode').asString;

      cdsOrderItem.fieldbyname('prodname').asString  := cdsCart.fieldbyname('prodname').asString;

      cdsOrderItem.fieldbyname('saleprice').asString := cdsCart.fieldbyname('saleprice').asString;

      cdsOrderItem.fieldbyname('qty').asString       := cdsCart.fieldbyname('qty').asString;

      cdsOrderItem.fieldbyname('amount').AsString    := cdsCart.fieldbyname('amount').AsString;

      cdsOrderItem.post;

      iSum := iSum + cdsOrderItem.fieldbyname('amount').Asinteger;

      iQty := iQty + cdsOrderItem.fieldbyname('qty').Asinteger;

      cdsCart.Delete;  //저장된건 삭제

    end;

    cdsCart.SaveToFile('./cart.xml');

    cdsOrderItem.SaveToFile('./orderitem.xml');

    // 주문서 저장

    cdsOrder.Append;

    cdsOrder.fieldbyname('memberid').asString     := UserSession.memberid;

    cdsOrder.fieldbyname('orderno').asString      := sOrderNo;

    cdsOrder.fieldbyname('orderdate').asdatetime  := date;

    cdsOrder.fieldbyname('totalamount').asinteger := iSum;

    cdsOrder.fieldbyname('totalqty').asinteger    := iqty;

    cdsOrder.fieldbyname('AccountState').asString := '1';

    cdsOrder.fieldbyname('deliveryState').asString:= '1';

    cdsOrder.fieldbyname('shipaddress').asString  := eAddress.Text;

    cdsOrder.fieldbyname('shippost').asString     := ePost.text;

    cdsOrder.fieldbyname('shiptel').asString      := eTel.Text;

    cdsOrder.fieldbyname('Recipient').asString    := eRecipient.Text;

    cdsOrder.fieldbyname('shipmemo').AsString     := eMemo.Text;

    cdsOrder.post;

    cdsOrder.SaveToFile('./order.xml');

    WebApplication.ShowMessage('주문이 처리되었습니다.' + #13#10 +

                               '주문번호는 ' + sOrderNo + ' 입니다.' + #13#10 +

                               '자세한정보는 주문조회에서 확인하세요' );

  end;

  IWRegionOrder.Visible := false;

end;
10.주문조회

주문조회는 마스터 – 디테일구조로 지금까지 주문한 내역을 그리드로 볼수있도록 구현하였다.

11.관리자기능

관리자는 관리자만 이동할수 있게 메뉴에서 작업을 했었다.

관리자기능은 크게 회원관리,상품관리,주문내역관리이다.

이 부분은 게시판과 같은 기능이라서 첨부한 소스만 참고 하면 될것같다..

이상으로 인트라웹을 이용한 쇼핑몰을 구축해보았다. 

인트라웹을 구현해보면서 웹어플리케이션의 강력함을 보았을 것이다. 이런 기술을 기초로 하면 어떤 웹솔루션도 쉽고 빠르게 구축할수 있고 다양한 플랫폼에서도 다 적용이 될수 있다. 그중에서도 카일릭스의 인트라웹은 리눅스 웹서버 환경에 아주 유용한 RAD툴에서 웹RAD툴로서 도움을 주고있다. 앞으로 많은 개발자들이 카일릭스의 인트라웹을 통해 좋은 웹 솔루션을 구축해주길 바란다.




















































